

# Attaching a PS/2 Keyboard

This document describes a method for attaching a PS/2 keyboard to the EVB001 evaluation board. The whole process from the hardware to arrayFORTH till polyFORTH is covered by this app note.

The text assumes you have familiarized yourself with our hardware and software technology by reading our other documents on those topics. The current editions of all GreenArrays documents, including this one, may be found on our website at <http://www.greenarraychips.com>. It is always advisable to ensure that you are using the latest documents before starting work

The work described herein is that of Stefan Mauerhofer.

## Contents

<b>1.</b>	<b>Introduction .....</b>	<b>2</b>
<b>2.</b>	<b>The PS/2 Interface .....</b>	<b>2</b>
<b>2.1</b>	<i>Physical Attachment.....</i>	<b>2</b>
<b>2.2</b>	<i>Interface Circuit.....</i>	<b>3</b>
<b>2.3</b>	<i>General Mode of Operation.....</i>	<b>4</b>
<b>2.3.1</b>	Receiving Data .....	<b>5</b>
<b>2.3.2</b>	Sending Data .....	<b>6</b>
<b>3.</b>	<b>arrayFORTH Implementation .....</b>	<b>7</b>
<b>3.1</b>	<i>The Data Line .....</i>	<b>7</b>
<b>3.2</b>	<i>The Clock Line .....</i>	<b>8</b>
<b>3.3</b>	<i>The Data Buffer .....</i>	<b>9</b>
<b>3.4</b>	<i>Send Buffer.....</i>	<b>10</b>
<b>3.5</b>	<i>Loading .....</i>	<b>10</b>
<b>3.6</b>	<i>Initialization .....</i>	<b>10</b>
<b>3.7</b>	<i>Receiving Data .....</i>	<b>11</b>
<b>3.8</b>	<i>Sending Data.....</i>	<b>12</b>
<b>3.9</b>	<i>SoftSim Integration .....</i>	<b>12</b>
<b>3.10</b>	<i>Testing on the Chip.....</i>	<b>13</b>
<b>4.</b>	<b>polyFORTH Implementation.....</b>	<b>14</b>
<b>4.1</b>	<i>Integrating the Keyboard Code.....</i>	<b>14</b>
<b>4.2</b>	<i>Accessing the PS/2 nodes from polyFORTH.....</i>	<b>15</b>
<b>4.3</b>	<i>Integration into polyFORTH.....</i>	<b>16</b>
<b>4.3.1</b>	Accessing GA144 Nodes.....	<b>16</b>
<b>4.3.2</b>	Buffers and Meta-Key Handling.....	<b>17</b>
<b>4.3.3</b>	Key Types and Keymap Table.....	<b>18</b>
<b>4.4</b>	<i>Using a polyFORTH Background Task .....</i>	<b>21</b>
<b>5.</b>	<b>Conclusion .....</b>	<b>23</b>
<b>6.</b>	<b>References .....</b>	<b>23</b>

## 1. Introduction

This application describes a way to attach a PS/2 keyboard to the EVB001 evaluation board using a small interface for level shifting between 1.8 and 5 volt circuitry. 4 nodes are used to drive the interface and sending/receiving bytes. The more complex protocol for sending bytes and waiting for acknowledgement and interpreting the scan codes is handled in the polyFORTH virtual machine.

## 2. The PS/2 Interface

PS/2 is a standard for attaching devices, mostly input devices like keyboard or mouse, to a computer. A good description of the line operations can be found in [1].

### 2.1 Physical Attachment

The PS/2 interfaces can be identified by the 6-pin mini-DIN socket. It uses a serial 2-wire bidirectional protocol for transmitting bytes.




Fig. 1 Mini-DIN PS/2 connector

Both lines (clock and data) are open-collector. That means they are low active and must be connected to 5V with a pull-up resistor, which allows communication in both directions.

The PS/2 interface implements a bidirectional synchronous serial protocol. The bus is “idle” when both lines are high (open-collector or open-drain). This is the only state where the keyboard/mouse is allowed begin transmitting data. The host has ultimate control over the bus and may inhibit communication at any time by pulling the clock line low.

## 2.2 Interface Circuit

The circuit for attaching a PS/2 device is quite simple. A TXS0108E open-drain level-shifter chip must be connected, because the PS/2 interface operates at 5 volts. Figure 2 shows the circuit:





Fig. 2 Interface circuit

No pull-up resistor is necessary, because it is already built into the chip. For details please consult the datasheet from Texas Instruments. The two decoupling capacitors can be neglected for a test environment, because all frequencies are below 20 kHz, which lies way below the limits of the level-shifter chip and the GA144. Picture 1 displays the experimental circuitry used for this application note.



Pic. 1 Test circuit

Picture 1 displays the test buildup. The level-shifter chip works fine in this configuration even without the 2 decoupling capacitors, although I recommend using them.

## 2.3 General Mode of Operation

When the interface is inactive, then both lines are at 5V. Normally the lines are controlled by the keyboard and the clock frequency ranges from 10 to 18 kHz. Figure 3 shows the timing, when the keyboard is sending a byte:




Fig. 3 Read operation and general timing

The bit sequence is: start, 8 x data (least significant bit first), parity (odd), stop.

The timing limits are:

$$a > 50\mu s$$

$$30\mu s < b, c < 50\mu s$$


$$d \approx 10-20\mu s$$

The keyboard controls both lines, so the GA144 should set the corresponding pins to high impedance input. Initially both lines (Data and Clock) are pulled to +5V (high).

**Note:** The clock pulses (for read and write!) are always generated by the keyboard. This makes integration into a GA144 easier, because no timer is needed.

### 2.3.1 Receiving Data

If the keyboard wants to send a byte it sets the data line low (start bit) followed by 8 data bits (LSB first), parity bit (odd) and a stop bit (high). The data line is changed during the high phase of the clock and stable during low phase. Picture 2 displays a screenshot from a logic analyzer's display while reading the byte 29h from the keyboard.



Pic. 2 Logic analyzer display while reading a byte from the keyboard

### 2.3.2 Sending Data

Because the lines are open-collector, it is possible to send a byte to the keyboard. The timing for this case is shown in figure 4:





Fig. 4 Write timing

Sending data to the keyboard is little more complicated than receiving. First the host sets clock low for at least 100 microseconds ( $100 \mu\text{s} < e < 10 \text{ ms}$ ). Then it sets the data line low, signaling that it wants to send data. Now the clock line is released. The host waits until the keyboard takes over and begins clocking. The data line is altered by the host during low clock phases and it holds the data while the clock is high. Note that is the inverse behavior compared to reading. The host releases the data line when the stop bit is reached. The keyboard then sends an acknowledgment bit for finishing the operation. Picture 3 displays the logic analyzer display of a keyboard reset sequence where FFh (reset) is sent to the keyboard that responds with FAh (ack).



Pic. 3 Sending reset (FFh) to keyboard with response (FAh)

Note the relative large delay (2 ms) from the clock release till the keyboard starts clocking. This is not a problem since the GA144 does not measure any time but waits for events to occur.

### 3. arrayFORTH Implementation

4 nodes are used to implement the basic serial protocol in F18 code. Fortunately the keyboard does generate the clock signal for reading and writing, so the nodes need no timed mechanism for generating a clock. Node 317 is used for the clock line and node 217 for the data line. Node 316 is a 1 entry sending buffer and node 216 a receiving 32 element FIFO buffer.

#### 3.1 The Data Line

The data line is handled by node 217. Listing 1 is the code for this node:

```
ps/2 interface driver - data,
connect data to pin 17 @ node 217
buffer bits to send or receive
bit current bit mask
start high and jump to down port
clear write n @ 0 and 1 @ 1
bo write next bit
5us wait approx. 5 microseconds
high put data line to high impedance
low drive data line to ground
bi read next bit
r1 read a 1 bit
next-bit set bit mask to next bit
parity nondestructive calculation of even parity
with output 0 or hex 100
byte-out read 1 byte from node 317 and calculates
odd parity and stop bit
accept finish reading of a byte
error parity error detected
ok pass received byte to next node
ack read acknowledge bit
```

```
920 list
ps/2 interface driver - data,
reclaim 217 node 0 org
buffer 00 0 ,*bit 01 0 ,
start 02 leap -d-- ;
clear 04 dup or dup a! !+ a ! ;
bo 07
5us 07 2000 for unext,
dup or a! @+ @ dup 2* ! and if
high 0d swap then dup or !b ; then
low 0e 20000 !b ;
bi 10 dup or a! @b -if
r1 12 @+ @ or dup dup or a! !,
then*next-bit 15 @+ @ 2* ! ;
parity 17 100 a! dup dup or over 7 for pn,
2* dup a and push over pop or over next drop ;
byte-out 21 down a! @ dup push parity,
pop or dup dup or a! 300 or !+ a ! ;
accept 29 dup or a! @ parity or 100 and if
error 2e ; then
ok 2e dup or a! @ ff and right a! ! ;
ack 33 high ;
```

Listing 1 Data line

After starting the node waits on the down port for code from node 317. The port executes directly any cell received from the clock node 317. Node 217 is offering a set of entry points that can be called by node 317.

## 3.2 The Clock Line

Node 317 is responsible for the clock line. It needs to place call instructions into the down port. The sequence `@p .. xxx ..` pushed the call instruction onto the stack. Since we didn't use **reclaim** we still have the word addresses compiled from block 920 and we can use them for block 922.

<pre> ps/2 interface driver - clock, , connect clock to pin 17 @ node 317 <b>wait0</b> wait until clock line is 0 <b>wait1</b> wait until clock line is 1 <b>20us</b> wait approx. 20 microseconds, this delay is critical for correct operation <b>waitr</b> wait for the clock going 1 and then 0 <b>da</b> set port a to clock node <b>from-ps/2</b> read a byte from ps/2 <b>stopbit</b> read and handling the stop bit <b>to-ps/2</b> write a byte to ps/2 <b>start</b> entry point of node <b>loop</b> poll the clock line or the port write bit to either begin a reading or a writing operation . this loop runs indefinitely leaving the node running . </pre>	<pre> <b>922 list</b> ps/2 interface driver - clock, 317 node 8 org <b>wait1</b> 08 dup dup or ahead <b>wait0</b> 0a 8000 then 0c !b left a! @ drop <b>da</b> 0f down a! ; <b>waitr</b> 11 wait1 leap wait0 then <b>20us</b> 14 8000 9000 10000 for unext ; <b>from-ps/2</b> 17 <b>da</b> @p .. clear .. ! 20us 1b, 9 for <b>waitr</b> @p .. bi .. ! next, <b>stopbit</b> 21 wait1 @p .. accept .. ! 20us ; <b>to-ps/2</b> 25 right a! @ 20000 !b da, @p .. byte-out .. ! ! 60000 for unext, startbit @p .. low .. ! 4000 for unext, dup or !b <b>wait1</b> 9 for 20us <b>wait0</b>, @p .. bo .. ! 20us next <b>waitr</b> 3d, @p .. ack .. ! <b>waitr</b> ; 40 0 org <b>start</b> 00 <b>loop</b> 00 @b - -if <b>from-ps/2</b> loop ; then 03, - 8000 and .. 05 if <b>to-ps/2</b> then <b>loop</b> : 08 </pre>
---	---

Listing 2 Clock line

The memory of the node is quite full. The main loop is spinning (thus burning around 4 mA) and waits either for a clock low pulse from the keyboard or an input from the right port. Once an operation (either sending or receiving) has started it will be completed without possibility of interruption. Since there are no timeouts used in the design it is not possible for the node to resynchronize its operation with the keyboard once a single bit has been added or is missing. Therefore the whole interface is not very robust regarding bit synchronization errors. However the stability is sufficient for an experimental environment.

There are 3 constants (10000, 60000 and 4000) that are important for the timing. It turns out that the timing parameter in the word **20us** is crucial. Once the node does lose a bit, its behavior is undefined and the chip must be reset. My keyboard is working without problem with values 9000 or 10000 but runs into sporadic error when using 8000.

Node 217 (data) and 317 (clock) work together very closely. This is an example of how to use the port execution facility of the GA144 in an efficient way.

### 3.3 The Data Buffer

Incoming data is buffered in a FIFO buffer in node 216. My first implementation was designed to block and wait for input from nodes 215 and 217 (left and right):

```
ps/2 interface driver - data buffer,
implementing a ringbuffer for 31 cells.
r! store read pointer
r@ fetch read pointer
w! store write pointer
w@ fetch write pointer
inc increment pointer and clip to buffer
-empty? test if buffer is not empty
-full? test if buffer is not full
start begin of program
loop processing loop . wait for command or data .
for a node to read a negative word must be send
read read from buffer
write write to buffer
```

```
924 list
ps/2 interface driver - data buffer,
reclaim 216 node 20 org
r! 20 @p drop !p ;*r@ 21 0 ;
w! 23 @p drop !p ;*w@ 24 0 ;
inc 26 n-n 1 . + 1f and ;
-empty? 2a -f w@*check? r@ or ;
-full? 2d -f w@ inc check? ;
start 30*loop 30 @b -if
read 31 -empty? if r@ dup inc r! a! @ dup then
drop !b loop ; then
write 39 -full? if drop w@ dup inc w! a! ! then loop
; 40
```

Listing 3 FIFO buffer (flawed design – not used)

There is a problem with this design. Both nodes 215 and 217 act totally independent from each other, thus it is possible that a node starts writing while the data from the other node is currently reading. In this case node 216 may read garbage and the behavior is undefined. The only way to solve this is to poll both sides continuously and to sacrifice some energy for keeping the node running. Greg Bailey gave this info to me as I was writing this app node and he reviewed my solution. This problem may be fixed in a future version of the chip.

```
ps/2 interface driver - data buffer,
implementing a ringbuffer for 15 cells.
r! store read pointer
r@ fetch read pointer
w! store write pointer
w@ fetch write pointer
inc increment pointer and clip to buffer
-empty? test if buffer is not empty
-full? test if buffer is not full
start begin of program
loop polling loop
```

```
926 list
ps/2 interface driver - data buffer,
reclaim 216 node 10 org
r! 10 @p drop !p ;*r@ 11 0 ;
w! 13 @p drop !p ;*w@ 14 0 15 ;
inc 16 n-n 1 . + f and ;
-empty? 1a -f w@*check? r@ or ;
-full? 1d -f w@ inc check? ;
start 20 dup !b drop
loop 21 a! @b and or if drop @ push -full? if
drop w@ a! pop ! w@ inc w! dup then then drop
drop,
a! @b and or .. if drop @ push -empty? if drop
r@ a! @ push r@ inc r! dup then drop a! pop !
loop ; then drop drop loop ; 3c
```

Listing 4 Fixed FIFO buffer using poll loop

Block 926 contains the looping variant, which works correctly in all circumstances. The preset circular stack (initialized in block 912) is used to feed the loop with the data it needs to function thus saving precious memory on the node leaving enough space for 16 values in the FIFO buffer.

### 3.4 Send Buffer

Node 316 contains a simple wire node for decoupling a polyFORTH write to keyboard from the current operation. In the case that the keyboard is sending a character, node 317 does not accept any data. A wire node is also always a 1-element Buffer. In this sense node 316 is used to prevent any writer from blocking.

```
928 list
ps/2 interface driver - decoupling wire node,
reclaim 316 node 0 org
start 00 dup or ~,
dup push dup push dup push dup push,
dup push dup push dup push dup push,
for begin @ !b unext unext 07
```

Listing 5 Wire node

### 3.5 Loading

A loading block is used to load all 4 nodes. Block 200 must contain the instruction **918 load**.

```
918 list
ps/2 loader,
reclaim,
920 load 922 load 926 load 928 load,
,reclaim
```

Listing 6 Loading block

### 3.6 Initialization

Block 912 is used for node initialization. This can be used for softsim or loading the code into the GA144 chip or integrate it into the polyFORTH boot stream.

```
912 list
ps/2 - code loader,
217 dup +node /ram 2 /p io /b,
317 dup +node /ram 0 /p io /b,
216 dup +node /ram 30 /p left right or 155 or /b,
316 dup +node /ram 0 /p left /a right /b,
```

Listing 7 Initialization block (early version - not used)

**/p** defines the starting point of the code. The expression **left right or 155 or /b** requires a little explanation. The goal is to initialize the register **b** with the combination of port **left** and **right**. Inside the F18-compiler we could use the predefined address **r-1-**. Here we don't have these words but we can use the port words. We can combine any ports by x-oring the port addresses. Since port addresses are x-ored with 155h we must add this operation if the number of port is even. We use 2 ports therefore we must x-or the result with 155h to get a correct port number.

This was my initial design. Since we cannot use the blocking code on node 216, we must use the polling code, which halves the available buffers cells from 31 to 15. But 16 cells are still enough for our purposes. The new code requests the stack to be filled with predefined values for saving node's memory and use the cycling stack. Listing 8 shows the modified initialization code.

```
912 list
ps/2 - code loader,
217 dup +node /ram 2 /p io /b,
317 dup +node /ram 0 /p io /b,
216 dup +node /ram 20 /p io /b left 0 800 left right
0 8000 right 201ff 9 /stack,
316 dup +node /ram 0 /p left /a right /b,
```

Listing 8 Initialization block with poll loop fix and stack prefill

### 3.7 Receiving Data

This and the next chapter explains how the clock node invokes calls to the data node and when.

First Node 317 waits for the clock line to go low. If this happens it starts a read cycle by calling [from-ps/2](#). Figure 5 shows what happens in this word:




Fig. 5 Node interaction for receiving a byte from the keyboard

The sequence **clear**, 10 x **bi** and **accept** is called from node 317 via the down port, timed by the clock line. Accept retrieves the byte and sends it to node 216. There it is buffered for later use e.g. by polyFORTH. Up to 15 bytes can be buffered in node 216 before it overflows and ignores any further data. To read a byte from node 216 any word must be sent to it from node 215. If an unread byte is available in the FIFO buffer, then it is returned otherwise the word sent from node 215 is returned, signaling a buffer empty condition.

### 3.8 Sending Data

Node 317 initiates sending data if it has been idle.




Fig. 6 Node interaction for sending a byte to the keyboard

The sequence **byte-out** followed by the data, **low**, 10 x **bo** and **ack** is called from node 317 via the down port, timed by the clock line. The clock line must be driven to low long enough that the keyboard does recognize a sent condition or aborts any sending operation. This waiting is done by **100us** in node 317. Node 217 calculates also the parity bit and sets the stop bit.

### 3.9 SoftSim Integration

If we want to test our interface with SoftSim then block 916 is used to define the environment.

```

916 list
ps/2 - softsim starter,
912 load
/clk softbed assign time @ 100 / 1 and 1 or ?v
p17v ! ;
317 !node /clk,
/dat softbed assign time @ 1300 / 1 and 1 or ?
v p17v ! ;
217 !node /dat,

```

Listing 9 SoftSim integration

A **916 load** must be placed in block 216 to include our example in SoftSim. The level at the clock line is changed every 100 ticks and on the data line every 1300 ticks. In order to see something we must change 3 timing constants in block 920 and 922, e.g. replace 40000 and with 4. After using SoftSim and before using the code on the chip we must undo all changes.

### 3.10 Testing on the Chip

For testing on the host chip we must have a block that loads the code onto the chip and another for retrieving the scan codes.

```
914 list
ps/2 - loader,
empty compile host load loader load,
0 708 hook 0 -hook,
912 load,
2 ship
```


Listing 10 Loading code onto the chip

First we type **914 load**. After some time the code is loaded into the chip and started. We can now type something into the keyboard, e.g. “test”. Now we must use **910 load** to reset the chip and display the FIFO buffer.

```
910 list
ps/2 - loader,
empty compile host load talk,
2 216 hook panel ?ram
```

Listing 11 Resetting the chip and displaying the FIFO buffer node

We should now see the following screen:



Pic. 4 Screenshot after typing “test” on the keyboard

We can now see the make- and break-codes the keyboard sends for every keystroke. **2c** is the make code for the key **t** and **f0 2c** is the break code for the same key. After reset most keyboards use the scan code set 2. See [2] for details.

## 4. polyFORTH Implementation

For using the full potential of the PS/2 interface, we can integrate it into a virtual machine that takes care of the higher level aspects. The larger memory of the polyFORTH virtual machine is better suited for configuring the keyboard and interpreting the scan codes.

### 4.1 Integrating the Keyboard Code

If we want to load our PS/2 code together with the polyFORTH virtual machine, then we can place a **912 load** into block 368 (or 478 for older versions), where the additional I/O for the virtual machine is loaded:

```
368 or 476 list
- additional i/o,
spi 705 +node 1606 /ram io /b a9 /p,
,
keybrd 912 load
```

Listing 12 Integrating our code into the polyFORTH boot stream

When polyFORTH is started then our code will also be loaded into nodes 216, 217, 316 and 317. That is true whether you start polyFORTH from the IDE (**450 load**) or you install the polyFORTH boot stream in the flash (**460 load**).

## 4.2 Accessing the PS/2 nodes from polyFORTH

After starting polyFORTH on the eval-board we must load the snorkel and ganglia mechanism to be able to access the keyboard nodes. **142 load** will do that. But before we load block 142 we must add the word **R!@** to it, if it is not already there (arrayFORTH version prior to h). Listing 13 displays the block 142 for arrayFORTH versions h or higher.

```

142
0   < General node access>  HEX  2VARIABLE fromRW
1 CREATE ~rw  :DOWN 2, 1 , 2033 , < pth> 0 W, < #io> 0 W, 4 W,
2   < a push @p @p> 2 , 2817 , HERE < n> 0 W, HERE < a> 0 W,
3   HERE < a! ! dup> 2 , 0BA92 , < !p pop a! ;> 0CCAD W,
4   HERE SWAP CONSTANT ~rwF  SWAP 1+ CONSTANT ~rwA
5   SWAP CONSTANT ~rwN
6 CREATE ~RW  :DOWN 2, o18 , ~rw - DUP . 2/ 1- W, ~rw W,
7   i18 , 0 W, fromRW W, HERE FIN CONSTANT ~RWfin
8 : ND < _> CREATE DOES R> 2@ ~rw 4 + 2! ;
9 ND N300 1 uu 7 11 0 dd ,path N300
10 ND N705 2 uu 2 11 1 uu ,path ND N600 2 uu 7 11 0 uu ,path
11 : r@! < df> ~rwF 2! ~RW ~RWfin +SNORK sDONE ;
12 : R! < d a> ~rwA ! ~rwN 2! < a! ! dup> 0BA92 2 r@! ;
13 : R@ < a - d> ~rwA ! < a! drop @> 0A20A 2 r@! fromRW 2@ ;
14 : R!@ < d a - d> ~rwA ! ~rwN 2! < a! ! @> 0BA0A 2 r@!
15   fromRW 2@ ;

```

Listing 13 Block 142 (arrayFORTH version h and later)

## 4.3 Integration into polyFORTH

Until now we have used our keyboard interface “by hand”. In order to fully support the keyboard and its features we must implement a little driver that converts the scan codes into ASCII code and keep track of meta-keys like shift and caps-lock. To simplify the task we assume that our keyboard supports the scan code set 3 (for details see [3]). Scan code set 3 is the simplest and most orthogonal set of the 3 scan code sets available. Be aware that some keyboards may not support scan code set 3. They will not work properly with the code provided in this app node.

There is a block (300) for loading the whole keyboard driver.

```

300
0 < keyboard block loader>
1 142 LOAD
2 301 308 THRU
3
4
5
6
7
8
9
10
11
12
13
14
15

```

Listing 14 Keyboard driver loading

### 4.3.1 Accessing GA144 Nodes

Block 301 defines the words (**keybr** and **keybw**) for actually accessing the PS/2 read and write nodes.

```

301
0 < keyboard interface>
1 ND n215 6 rr 1 dd 1 rr ,path
2 ND n315 7 rr 0 0 ,path
3
4 HEX 175 CONSTANT left DECIMAL
5
6 : keybw < c> n315 0 left R! ;
7 : keybr < - d> n215 -1 -1 left R@P ;
8 : tstr 9 FOR keybr D. NEXT ;
9
10
11
12
13
14
15

```

Listing 15 Accessing PS/2 nodes

Users of arrayFORTH version g or earlier must be aware that the snorkel/ganglia node is different from version h and higher (node 108 if version <= g else node 207). The path used in Listing 15 is for arrayFORTH version h or higher.

**keybw** writes a byte to the left port of node 315, thus writing to node 316. Node 316 is a wire node and reads the byte, if it is empty. Node 316 is basically not necessary but it decouples the polyFORTH virtual machine from node 317, which might be busy.

**keybr** reads the 15 byte FIFO buffer in node 216, accessing it via node 215. If the FIFO is empty then **keybr** returns -1. Otherwise the next byte is removed from the FIFO and **b 0** is returned.

The word **tstr** is used to read and display 10 values from the buffer in node 216. It is convenient for testing.

### 4.3.2 Buffers and Meta-Key Handling

Block 302 contains words for handling 3 buffers: **kbd-i**, **kbd-o** and **kbd-b**. Each buffer contains 16 cells organized as a FIFO queue. An array with 6 elements contains the read and write indexes for the buffers. Buffers **-i** and **-o** are the raw input and output buffers, while buffer **-b** holds the ASCII keys after scan code translation.

```

302
0 < keyboard driver>
1 48 ARRAY kbd-buf 6 ARRAY kbd-idx
2 0 CONSTANT kbd-i 2 CONSTANT kbd-o 4 CONSTANT kbd-b
3 16 2 kbd-idx ! 16 3 kbd-idx ! 32 4 kbd-idx ! 32 5 kbd-idx !
4 : kbd++ < n - n'> DUP 48 AND SWAP 1+ 15 AND OR ;
5 : kbd-i+ < i -> DUP kbd-idx @ kbd++ SWAP kbd-idx ! ;
6 : ?kbd-empty < i - f> DUP 1+ kbd-idx @ SWAP kbd-idx @ = ;
7 : ?kbd-full < i - f> DUP 1+ kbd-idx @ kbd++ SWAP kbd-idx @ = ;
8 : kbd-w < c i -> DUP ?kbd-full IF DROP DROP ELSE 1+ SWAP OVER
9   kbd-idx @ kbd-buf ! kbd-i+ THEN ;
10 : kbd-r < i - c> DUP ?kbd-empty IF DROP 0 ELSE DUP kbd-idx @
11   kbd-buf @ SWAP kbd-i+ THEN ;
12 : kbd-flush < -> 0 kbd-idx 0 OVER ! 1+ 0 OVER ! 1+ 16 OVER !
13   1+ 16 OVER ! 1+ 32 OVER ! 1+ 32 SWAP ! ;
14 : kbd-wait < - c> BEGIN keybr WHILE DROP PAUSE REPEAT ;
15 : kbd-read < -> kbd-wait kbd-i kbd-w ;

```

Listing 16 Buffer and buffer management words

The word **kbd++** increments an index and limits it to the buffer it points to. An index is incremented with the word **kbd-i+**. The words **?kbd-empty** and **?kbd-full** are used for testing whether a buffer is empty or full. **kbd-w** writes a cell into a buffer, e.g. **2 kbd-i kbd-w** writes 2 into the raw input buffer. **kbd-r** reads the next cell from a buffer or returns 0 if the buffer was empty. The word **kbd-flush** clears all 3 buffers. **kbd-wait** waits for the next byte from the PS/2 interface and **kbd-read** puts the next byte from the PS/2 interface into the raw input buffer.

A keyboard contains certain keys that do not generate an ASCII code but have an influence on the interpretation of other keys, e.g. shift or caps-lock. These meta-keys must be specially treated. Block 303 contains the constant definitions for these meta-keys and the variable **kbd-meta**, holding the state of the meta-keys.

```
303
0 < keyboard driver>
1 : kbd-wr < r c -> keybw BEGIN kbd-wait OWER OWER XOR WHILE
2   kbd-i kbd-w REPEAT DROP DROP ; HEX
3 : kbd-reset < -> kbd-flush AA FF kbd-wr FA F0 kbd-wr
4   FA 3 kbd-wr FA FA kbd-wr kbd-flush ;
5 : kbd-led < n -> ED kbd-o kbd-w ? AND kbd-o kbd-w ;
6 VARIABLE kbd-meta 0 kbd-meta !
7 0001 CONSTANT KYSK 0002 CONSTANT KYNK
8 0004 CONSTANT KYCK 0008 CONSTANT KYFN
9 1000 CONSTANT KYLS 2000 CONSTANT KYRS
10 4000 CONSTANT KYLA 8000 CONSTANT KYRA
11 0100 CONSTANT KYLC 0200 CONSTANT KYRC
12 0400 CONSTANT KYLM 0800 CONSTANT KYRM
13 0010 CONSTANT KYLV 0020 CONSTANT KYRW
14 0040 CONSTANT KYAG 0080 CONSTANT KYMN
15 DECIMAL
```

## Listing 17 Meta key constants

Apart from the meta-key definitions 3 useful words are defined. First **kbd-wr** is used to send a byte to the PS/2 interface and to wait for a specific reply. The word **kbd-reset** resets the keyboard and configure it to use scan code set 3. With **kbd-led** the leds on the keyboard can be set. The state of a meta-key is defined by a bit in the variable **kbd-meta**. The bit numbers are defined so that the lower 3 bits can be directly used for setting the keyboard leds.

### 4.3.3 Key Types and Keypad Table

All keys can be grouped to key types with similar behavior, e.g. all alpha keys are sensitive to the state of the caps-lock and the shift-keys. Blocks 304 to 306 contain the words to define all needed key types.

```
304
0 < keyboard_driver>
1 : ?kbd-shift < -f> kbd-meta @ [ KYLS KYRS OR ] ;
2   LITERAL AND @= NOT ;
3 : ?kbd-ashift < -f> ?kbd-shift kbd-meta @ KYCK AND
4   IF NOT THEN ;
5 : ?kbd-alt < -f> kbd-meta @ [ KYLA KYRA OR ] ;
6   LITERAL AND @= NOT ;
7 : ?kbd-ctrl < -f> kbd-meta @ [ KYLC KYRC OR ] ;
8   LITERAL AND @= NOT ;
9 : kbd+f1 < add flags> 255 AND kbd-meta @ 65280 AND OR ;
10 : kbdi < ignore> DROP -1 ;
11 : kbda < alpha> ?kbd-ashift IF Hi ELSE Lo THEN kbd+f1 ;
12 : kbds < shift> ?kbd-shift IF Hi ELSE Lo THEN kbd+f1 ;
13 : kbdn < num> kbd-meta @ KYNK AND @= IF Lo ELSE Hi THEN
14   kbd+f1 ;
15 : kbdt < toggle> kbd-meta @ XOR kbd-meta ! -1 ;
```

### **Listing 18 Meta key words**

The words **?kbd-shift** and **?kbd-ashift** are used for determining whether the basic value or the shifted value of a key must be taken. **kbd+f1** adds some meta-key flags to a character byte. **kbdi** is used when a key must be ignored. Most break codes can be ignored. **kbda** is used for alpha keys, **kbds** for shifted. Key on the numerical keybad must use **kbdn** and meta-key toggles (e.g. caps-lock) use **kbdt**.

```

305
0 < keyboard driver>
1 24 ARRAY kbd-fun < function keys>
2 : khdf < function key> ?kbd-shift IF Hi ELSE Lo THEN
3   kbd-fun @ ?DUP IF EXECUTE THEN ;
4 : khd-uled < update led> kbd-meta @ ? AND kbd-led kbdi ;
5 : khdm < meta make> kbd-meta @ OR kbd-meta ! ;
6 : khdb < meta brk> INUERT kbd-meta @ AND kbd-meta ! ;
7 VARIABLE kbd-brk @ kbd-brk !
8 : kbr < reset break> @ kbd-brk ! ;
9 : khi < ignore> kbdi kbr ;
10 : khk < set break> -1 kbd-brk ! kbdi ;
11 : kba < alpha> kbd-brk @ IF kbi ELSE kbda THEN ;
12 : kbs < shift> kbd-brk @ IF kbi ELSE kbds THEN ;
13 : kf < func> kbd-brk @ IF kbi ELSE kbdf THEN ;
14 : kbx < fix> kbd-brk @ IF kbi ELSE Lo kbd+f1 THEN ;
15 : kbm < meta> kbd-brk @ IF kbdb kbr ELSE kbdm THEN ;

```

Listing 19 Function keys and break code flag

**kbd-fun** is an array of function key vectors. Index 0 to 11 are used for the function keys F1 to F12 while index 12 to 23 for the shifted function keys.

The word **kbdf** is used for function key, **kbd-uled** updates the keyboard leds and **kbdm** is used for meta-keys like shift. The variable **kbd-brk** holds a flag that is set when a break code (F0) is received from the keyboard, which informs the driver to ignore the next scan code.

The 3 character words **kb\_** handle the corresponding make and break codes.

```


306
0 < keyboard driver>
1 : kbn < num> kbd-brk @ IF kbi ELSE kbdi THEN ;
2 : kbt < toggle> kbd-brk @ IF kbi ELSE kbdt kbd-uled THEN ;
3 VARIABLE kbd-last @ kbd-last !
4 : kbd < resend> kbd-last @ keybw kbi ;
5 : kbc < ack> @ kbd-last ! kbi ;
6 : kbd-snd < -> kbd-last @ @= IF kbd-o ?kbd-empty NOT IF
7   kbd-o kbd-r DUP kbd-last ! keybw THEN THEN ;
8 VARIABLE kbd-map
9 : kbd-in keybw IF DROP ELSE kbd-i kbd-w THEN ;
10 : !kbd < n -> DUP -1 = IF DROP ELSE kbd-b kbd-w THEN ;
11 : kbd-rcv < -> kbd-i ?kbd-empty NOT IF kbd-b ?kbd-full NOT IF
12   kbd-i kbd-r 2* kbd-map @ + DUP @ SWAP 1+ @ EXECUTE
13   !kbd THEN THEN ;
14
15

```

Listing 20 Last byte send and keyboard processing words

The variable **kbd-last** holds the last byte sent to the keyboard. If it is not 0 then an acknowledgment from the keyboard is pending. It is also used to resend the byte when the keyboard replies with FE. The variable **kbd-map** contains the address of the current key translation table. **kbd-snd** and **kbd-rcv** are used to process the raw input and output buffers.

Block 307 contains the final 2 character words for compiling the key definitions.



```


307
0 < keyboard driver>
1 : KI < n -> , [ ' kbi ] LITERAL , ;
2 : KA < n -> , [ ' kba ] LITERAL , ;
3 : KM < n -> , [ ' kbm ] LITERAL , ;
4 : KF < n -> , [ ' kbf ] LITERAL , ;
5 : KS < n -> , [ ' kbs ] LITERAL , ;
6 : KX < n -> , [ ' kbx ] LITERAL , ;
7 : KN < n -> , [ ' kbn ] LITERAL , ;
8 : KC < n -> , [ ' kbc ] LITERAL , ;
9 : KK < n -> , [ ' kbk ] LITERAL , ;
10 : KD < n -> , [ ' kbd ] LITERAL , ;
11 : KT < n -> , [ ' kbt ] LITERAL , ;
12
13 0 TABLE keyb-us 310 HEX LOAD DECIMAL 311 HEX LOAD DECIMAL
14 keyb-us kbd-map ! < install key map>
15

```

Listing 21 2 character key compiling words

These 2 character short words are used to define the keys in the key map, which is loaded from block 310 and 311. The key types handled are ignore (KI), alpha (KA), meta (KM), function (KF), shift (KS), fix (KX), numeric (KN), acknowledge (KC), resend (KD) and toggle (KT).

Block 310 contains the first half of the key map. It is used by **kbd-rcv** to convert scan codes to ASCII codes or to handle special keys. Even the handling of an acknowledgment (FA) or resend (FE) code from the keyboard is defined by this table.



```

310
0 0000 KI 0D01 KF
1 001B KX 0000 KI 0000 KI 0000 KI 0000 KI 0000 KI 0009 KX 7E60 KS 0E02 KF
2 0000 KI KYLC KM KYLS KM 0000 KI KYCK KT 5171 KA 2131 KS 0F03 KF
3 0000 KI KYLA KM 5A7A KA 5373 KA 4161 KA 5777 KA 4032 KA 1004 KF
4 0000 KI 4363 KA 5878 KA 4464 KA 4565 KA 2434 KS 2333 KS 1105 KF
5 0000 KI 0020 KX 5676 KA 4666 KA 5474 KA 5272 KA 2535 KS 1206 KF
6 0000 KI 4E6E KA 4262 KA 4868 KA 4767 KA 5979 KA 5E36 KS 1307 KF
7 0000 KI KYRA KM 4D6D KA 4A6A KA 5575 KA 2637 KS 2A38 KS 1408 KF
8 0000 KI 3C2C KS 4B6B KA 4969 KA 4F6F KA 2930 KS 2839 KS 1509 KF
9 4969 KA 3E2E KS 3F2F KS 4C6C KA 3A3B KS 5070 KA 5F2D KS 160A KF
10 0000 KI 0000 KI 2227 KS 0000 KI 7B5B KS 2B3D KS 170B KF 0000 KI
11 KYRC KM KYRS KM 000D KX 7D5D KS 7C5C KS 0000 KI 180C KF 0000 KI
12 0000 KI 0000 KI 0000 KI 0000 KI 007F KX 0000 KI 0008 KX 0000 KI
13 0000 KI 3100 KN 0000 KI 3400 KN 3700 KN 0000 KI 0000 KI 0000 KI
14 3000 KN 2E7F KN 3200 KN 3535 KN 3600 KN 3800 KN KYNK KT 002F KX
15 0000 KI 0D0D KN 3300 KN 0000 RI 2B2B RN 3900 KN 2A2A KN 0000 KI

```

Listing 22 US-keyboard map for scan codes 00 .. 7F

Block 311 contains the upper half of the key map.

```
311
0 0000 KI 0000 KI 0000 KI 0000 KI 002D KX 0000 KI 0000 KI 0000 KI
1 0000 KI 0000 KI 0000 KI KYLW KM KYRW KM KYMN KM 0000 KI 0000 KI
2 0000 KI 0000 KI
3 0000 KI 0000 KI
4 0000 KI 0000 KI
5 0000 KI 0000 KI
6 0000 KI 0000 KI
7 0000 KI 0000 KI
8 0000 KI 0000 KI
9 0000 KI 0000 KI
10 0000 KI 0000 KI
11 0000 KI 0000 KI
12 0000 KI 0000 KI
13 0000 KI 0000 KI
14 0000 KK 0000 KI 0000 KI
15 0000 KI 0000 KI 0000 KC 0000 KI 0000 KI 0000 KI 0000 KI 0000 KD 0000 KI
```

Listing 23 US-keyboard map for scan codes 80 .. FF

The key map used in this document is for a US-layout keyboard. For keyboards with other layouts the key map must be adapted and the keyboard driver eventually adapted.

#### 4.4 Using a polyFORTH Background Task

Finally we can put all pieces of the driver together and integrate the keyboard driver as a background task into polyFORTH. Line 1 in block 308 defines a background task. With **+kbd** the task is activated and with **-kbd** it is stopped.


```
308
0 < keyboard background task for processing input>
1 64 64 BACKGROUND kbd-task
2 : +kbd kbd-task ACTIVATE kbd-reset BEGIN
3   kbd-in kbd-rcv kbd-snd PAUSE
4   AGAIN ; RECOUER
5 : -kbd kbd-task ACTIVATE NOD ;
6
7 : @kbd < -n> kbd-b kbd-r ;
8
9 : kbd-test BEGIN @kbd ?DUP IF 255 AND DUP 32 = IF DROP EXIT THEN
10   EMIT THEN PAUSE AGAIN ;
11
12
13
14
15
```

Listing 24 polyFORTH background task

The word **kbd-test** is a small test program. It loops and emits every key entered until the space bar is hit. It depends on the background task for running. Note that block 308 will not load for arrayFORTH versions prior to h.

More details about multitasking can be found in [4] and [5].

Block 309 is not loaded though block 300, because it replaces the terminal '**KEY**' vector and switches any keyboard input to the PS/2 interface. Block access via the serial terminal is not possible while using the PS/2 keyboard. **kbd-detach** will restore the old vector, thus returning to the serial terminal for input.



```

309
0 < keyboard replacement>
1 : kbd->key kbd-b ?kbd-empty IF 0 ELSE kbd-b kbd-r 255 AND THEN ;
2
3 : kbd-key BEGIN kbd->key ?DUP IF EXIT THEN PAUSE AGAIN ;
4
5 VARIABLE kbd-key-old
6
7 : kbd-attach 'KEY @ kbd-key-old ! [ ' kbd-key ] LITERAL 'KEY ! ;
8
9 : kbd-detach kbd-key-old @ 'KEY ! ;
10
11 +kbd
12 kbd-attach
13
14
15

```

Listing 25 Keyboard takeover

This driver is only an example of how to access a keyboard. If a software needs also the break codes from the keyboard, then the driver must be modified. If you have a keyboard with more keys than a standard us-layout, then you can very easily detect the scan codes (**tstr**) and modify the key map for these new keys.

Function keys can be attached easily to words (e.g. '**xxx 0 kbd-fun !**' will attach the word xxx to the F1 key).

This driver only generates ASCII values used by the word **KEY**. It requires 4 nodes on the GA144 and about 2k cells (including 512 cells for the key map) in the polyFORTH dictionary. The buffers are managed by a polyFORTH background task.

## 5. Conclusion

First we demonstrated how to use the multi-computer paradigm of the GA144 chip to access a PS/2 device. Then we integrated the higher level aspects for handling a keyboard into the polyFORTH environment. The combination of nodes allocated for a specific task and the flexibility of the polyFORTH virtual machine gives the developer a high degree of freedom to design hardware and software.

The author of this document is a C++ programmer and began using the arrayFORTH environment and the EVB001 evaluation board about 6 months prior to this app note as a hobbyist. With the help of the documents and staff from GreenArrays it was possible to write the driver software in a few weeks. Although arrayFORTH and polyFORTH are very different than environments most programmers are used to, it is worthwhile to begin working with these GreenArrays tools.

This application note was finished during a vacation visit to GreenArrays in Incline Village, NV. The author wants to thank for the help and patience he received from the GreenArray staff.

## 6. References

- [1] <http://www.computer-engineering.org>
- [2] <http://www.computer-engineering.org/ps2keyboard/>
- [3] <http://www.computer-engineering.org/ps2keyboard/scancodes3.html>
- [4] “polyFORTH Reference manual”, GreenArrays Data Books: DB005
- [5] “G144A12 polyFORTH supplement to DB005”, GreenArrays Data Books: DB06
- [6] “F18A Technology Reference”, GreenArrays Data Books: DB001

### IMPORTANT NOTICE

GreenArrays Incorporated (GAI) reserves the right to make corrections, modifications, enhancements, improvements, and other changes to its products and services at any time and to discontinue any product or service without notice. Customers should obtain the latest relevant information before placing orders and should verify that such information is current and complete. All products are sold subject to GAI's terms and conditions of sale supplied at the time of order acknowledgment.

GAI disclaims any express or implied warranty relating to the sale and/or use of GAI products, including liability or warranties relating to fitness for a particular purpose, merchantability, or infringement of any patent, copyright, or other intellectual property right.

GAI assumes no liability for applications assistance or customer product design. Customers are responsible for their products and applications using GAI components. To minimize the risks associated with customer products and applications, customers should provide adequate design and operating safeguards.

GAI does not warrant or represent that any license, either express or implied, is granted under any GAI patent right, copyright, mask work right, or other GAI intellectual property right relating to any combination, machine, or process in which GAI products or services are used. Information published by GAI regarding third-party products or services does not constitute a license from GAI to use such products or services or a warranty or endorsement thereof. Use of such information may require a license from a third party under the patents or other intellectual property of the third party, or a license from GAI under the patents or other intellectual property of GAI.

Reproduction of GAI information in GAI data books or data sheets is permissible only if reproduction is without alteration and is accompanied by all associated warranties, conditions, limitations, and notices. Reproduction of this information with alteration is an unfair and deceptive business practice. GAI is not responsible or liable for such altered documentation. Information of third parties may be subject to additional restrictions.

Resale of GAI products or services with statements different from or beyond the parameters stated by GAI for that product or service voids all express and any implied warranties for the associated GAI product or service and is an unfair and deceptive business practice. GAI is not responsible or liable for any such statements.

GAI products are not authorized for use in safety-critical applications (such as life support) where a failure of the GAI product would reasonably be expected to cause severe personal injury or death, unless officers of the parties have executed an agreement specifically governing such use. Buyers represent that they have all necessary expertise in the safety and regulatory ramifications of their applications, and acknowledge and agree that they are solely responsible for all legal, regulatory and safety-related requirements concerning their products and any use of GAI products in such safety-critical applications, notwithstanding any applications-related information or support that may be provided by GAI. Further, Buyers must fully indemnify GAI and its representatives against any damages arising out of the use of GAI products in such safety-critical applications.

GAI products are neither designed nor intended for use in military/aerospace applications or environments unless the GAI products are specifically designated by GAI as military-grade or "enhanced plastic." Only products designated by GAI as military-grade meet military specifications. Buyers acknowledge and agree that any such use of GAI products which GAI has not designated as military-grade is solely at the Buyer's risk, and that they are solely responsible for compliance with all legal and regulatory requirements in connection with such use.

GAI products are neither designed nor intended for use in automotive applications or environments unless the specific GAI products are designated by GAI as compliant with ISO/TS 16949 requirements. Buyers acknowledge and agree that, if they use any non-designated products in automotive applications, GAI will not be responsible for any failure to meet such requirements.

The following are trademarks or registered trademarks of GreenArrays, Inc., a Nevada Corporation: GreenArrays, GreenArray Chips, arrayForth, and the GreenArrays logo. polyFORTH is a registered trademark of FORTH, Inc. ([www.forth.com](http://www.forth.com)) and is used by permission. All other trademarks or registered trademarks are the property of their respective owners.

For current information on GreenArrays products and application solutions, see [www.GreenArrayChips.com](http://www.GreenArrayChips.com)

Mailing Address: GreenArrays, Inc., 774 Mays Blvd #10 PMB 320, Incline Village, Nevada 89451

Printed in the United States of America

Phone (775) 298-4748 fax (775) 548-8547 email [Sales@GreenArrayChips.com](mailto:Sales@GreenArrayChips.com)

Copyright © 2010-2012, GreenArrays, Incorporated

